Solutions to Exercise: Linear Depletion

PF@SuS@UniHD 2016

We assume that the junction is at $x=0$.
To the left, we have 'infinite' p-doping.
To the right, we assume a n-doping density following a power law n(x) = A x^{α}.

Constants, Definitions

In[52]:=
```
Clear[ρ, V, α, k, A];
$Assumptions = α ∈ Reals && k > 0 && A > 0;
```

In[54]:=
```
ρ[x_] = A x^α; (* Ansatz for doping profile *)
```

1. Field E(x)

In[55]:=
```
EField[x_] = Integrate[ρ[y], {y, 0, x}]
```

Out[55]= ConditionalExpression$\left[\frac{A\, x^{1+\alpha}}{1+\alpha}, \alpha > -1\right]$

In[56]:=
```
Simplify[EField[x], α > -1] // TeXForm
```

Out[56]//TeXForm=

```
\frac{A x^{\alpha +1}}{\alpha +1}
```

2: V[x] and V[T]

In[57]:=
```
V[x_] = Integrate[EField[y], {y, 0, x}]
```

Out[57]= ConditionalExpression$\left[\frac{A\, x^{2+\alpha}}{2+3\,\alpha+\alpha^2}, \alpha > -1\right]$

In[58]:=
```
V[T_, α_] = Simplify[V[T], α > -1]
```

Out[58]= $\frac{A\, T^{2+\alpha}}{2+3\,\alpha+\alpha^2}$

3. T[V]

In[59]:=
```
Solve[V[T, α] == V, T] // First
```

Solve: Inverse functions are being used by Solve, so some solutions may not be found; use Reduce for complete solution information.

Out[59]= $\left\{T \to \left(\frac{V\left(2+3\alpha+\alpha^2\right)}{A}\right)^{\frac{1}{2+\alpha}}\right\}$

In[60]:=
```
T[V_, α_] = T /. %
```

Out[60]= $\left(\frac{V\left(2+3\alpha+\alpha^2\right)}{A}\right)^{\frac{1}{2+\alpha}}$

In[61]:=
```
T[V, 0] (* Check the case of constant doping *)
```

Out[61]= $\sqrt{2}\sqrt{\frac{V}{A}}$

4: Constant doping

In[62]:=
```
T[V, α]
```

Out[62]= $\left(\frac{V\left(2+3\alpha+\alpha^2\right)}{A}\right)^{\frac{1}{2+\alpha}}$

In[63]:=
```
EQ1 = PowerExpand[T[V, α]] (* Manually extract the voltage dependence *)
```

Out[63]= $A^{-\frac{1}{2+\alpha}}\, V^{\frac{1}{2+\alpha}}\left(2+3\alpha+\alpha^2\right)^{\frac{1}{2+\alpha}}$

In[64]:= $\text{Solve}\left[V^{\frac{1}{2+\alpha}} == \sqrt{V}, \alpha\right]$ // First // Simplify

Solve: Inverse functions are being used by Solve, so some solutions may not be found; use Reduce for complete solution information.

Out[64]= $\{\alpha \to 0\}$

5: Exponent *α* required for T prop. to V

In[65]:= $\text{Solve}\left[V^{\frac{1}{2+\alpha}} == V, \alpha\right]$ // First // Simplify

Solve: Inverse functions are being used by Solve, so some solutions may not be found; use Reduce for complete solution information.

Out[65]= $\{\alpha \to -1\}$

In[66]:=
```
ρ[x] /. α → -1
```

Out[66]= $\frac{A}{x}$

In[67]:= `T[V, -1]`

Out[67]= 0

Additional: Try a ‘shifted’ 1/x doping without divergence at x=0. No solution found.

```
ρ1[x_] = A (x + os)^α;
$Assumptions = α ∈ Reals && k > 0 && A > 0 && os > 0 && x ∈ Reals;
(* add an offset, so that ρ[0] is finite *)
```

```
ρ1[0]
```

$A\, os^{\alpha}$

```
Plot[{ρ[x], ρ1[x]} /. {α → -1, A → 1, os → 1},
 {x, 0, 5}, PlotRange → {0, 2}, Frame → True]
(* Lower curce is new potential which is finite at x=0 *)
```




```
EField1[x_] = Integrate[ρ1[y], {y, 0, x}]
```

$$\text{ConditionalExpression}\left[\frac{A\left(-os^{1+\alpha}+(os+x)^{1+\alpha}\right)}{1+\alpha},\ x>0\right]$$

```
V1[x_] = Integrate[EField1[y], {y, 0, x}] // FullSimplify
```

$$\text{ConditionalExpression}\left[\begin{cases}\frac{A\left((os+x)^{2+\alpha}-os^{1+\alpha}\,(os+x\,(2+\alpha))\right)}{(1+\alpha)\,(2+\alpha)} & x>0\\ 0 & \text{True}\end{cases},\ x\geq 0\right]$$

```
Simplify[V1[T], T > 0]
```

$$\frac{A\left((os+T)^{2+\alpha}-os^{1+\alpha}\,(os+T\,(2+\alpha))\right)}{(1+\alpha)\,(2+\alpha)}$$

```
Limit[%, α → -1]
```

$-A\,(T+(os+T)\,\text{Log}[os]-(os+T)\,\text{Log}[os+T])$

```
Solve[% == V, T] // First
```

··· Solve: This system cannot be solved with the methods available to Solve.

$-A\,(T+(os+T)\,\text{Log}[os]-(os+T)\,\text{Log}[os+T]) == V$

$$\rho 2[x\_] = \begin{cases} A\, os^{\alpha} & x < os \\ A\, x^{\alpha} & \text{True} \end{cases}$$

$$\begin{cases} A\, os^{\alpha} & x < os \\ A\, x^{\alpha} & \text{True} \end{cases}$$

```
EField2[x_] = Integrate[ρ2[y], {y, 0, x}] // Simplify
```

$$\begin{cases} A\, os^{\alpha}\, x & os \geq x \,\&\&\, x \neq 0 \\ \frac{A\, \left(x^{1+\alpha}+os^{1+\alpha}\, \alpha\right)}{1+\alpha} & os < x \,\&\&\, x \geq 0 \\ 0 & \text{True} \end{cases}$$

```
V2[x_] = Integrate[EField2[y], {y, 0, x}] // FullSimplify
```

$$\begin{cases} \frac{1}{2}\, A\, os^{\alpha}\, x^2 & os \geq x \,\&\&\, x \neq 0 \\ \frac{2\, A\, x^{2+\alpha} - A\, os^{1+\alpha}\, \alpha\, (os\, (1+\alpha) - 2\, x\, (2+\alpha))}{2\, (1+\alpha)\, (2+\alpha)} & os < x \\ 0 & \text{True} \end{cases}$$